
"SÒCRATES:  Tu creus, doncs, que ell hauria estat disposat a investigar i aprendre una cosa que ell no
sabia, però que creia saber, abans d'haver-se sentit perplex per haver arribat a tenir  consciència de la seva
ignorància, i d'haver concebut el desig d'aprendre?
MENÓ: No crec que fos així, Sòcrates."

Menó o De la Virtu t, 85a
1. INTRODUCCIÓ

El programa Sòcrates és un entorn orientat a la resolució d'exercicis. Inclou
un mecanisme de seguiment dels exercicis (tant els que s'han resolt com els
que no) que permet avaluar el treball efectuat pels alumnes.
La facilitat de crear exercicis nous fa possible que el Sòcrates tingui tots els
avantatges d'un entorn obert. Cal dominar un llenguatge de definició
d'exercicis per poder-ne programar. Aquest llenguatge ve explicat en la
segona part d'aquest manual.
Llegiu abans el Manual de l'alumne per tal de comprendre les explicacions
que segueixen.

2. GESTIÓ DE L'APLICACIÓ
La gestió de l'aplicació requereix tenir coneixement de les sessions que hi ha
disponibles, el lloc on es troben i el grau de dificultat dels exercicis que les
componen, entre altres informacions. Seguidament es comenten els aspectes
del programa que el professor ha de conèixer.

Estructura dels directoris
Malgrat que l'aplicació pot funcionar des d'un disquet, es recomana executar-
la des de disc dur per motius de velocitat d'execució. Habitualment, el
directori on està situada l'aplicació és C:\SOCRATES.
Del directori de l'aplicació pengen un o més directoris de sessions. Cada un
d'ells conté sessions afins quant a temàtica, que poden compartir o no els
diversos exercicis. Dins d'aquests directoris de sessions, hi ha els directoris
d'exercicis. Cadascun d'ells pren el nom de l'exercici que conté, i inclou tots
els seus elements: fórmules, gràfics i el fitxer de l'exercici.
Aquesta estructura de directoris es pot variar, però aleshores s'han de
modificar els fitxers dels exercicis o de les sessions (en funció de quins



fitxers afecti el canvi) per poder funcionar correctament amb la nova
distribució.
Quan el professor assigna el treball que han de realitzar els alumnes, els ha
de comunicar la situació dins el disc de les sessions que vol que es treballin.
Del directori de l'aplicació també penja un directori anomenat TRAÇA, on
s'hi posen les traces dels diversos exercicis.
Per exemple, una estructura de directoris adequada per a l'aplicació podria
ser aquesta:

Noms d'arxius
En els diferents apartats, sovint caldrà especificar el nom d'un fitxer. Si el
nom no té camí (path ), l'aplicació el busca al directori actual que, segons el
cas, serà el de la sessió o el del problema. En cas que s'indiqui el camí, se
l'ha d'indicar complet; des de l'arrel del disc.
Hi ha la possibilitat, però, de començar el camí amb els caràcters ".\".
Aquests caràcters no signifiquen el directori actual (com passa en el sistema
operatiu) sinó el directori on es troba l'aplicació Sòcrates. Per exemple, en
l'estructura de directoris de l'exemple anterior, seria C:\SOCRATES..

Claus dels alumnes
Abans d'entrar en una sessió, l'aplicació en demana el nom i la clau a
l'alumne. Aquesta clau ha de ser un identificador. Per tant, s'ha de donar una
clau diferent a cada un dels alumnes. Aquesta clau pot tenir xifres i lletres,
però no caràcters especials (com poden ser lletres accentuades, signes de
puntuació, etc). La llargada de la clau ha de ser de vuit caràcters com a
màxim.

Fitxers de traça
Des de que l'alumne entra en un exercici fins que en surt, totes les seves
accions són memoritzades. En sortir de l'exercici, tant si l'ha resolt com si
no, aquesta traça queda desada en el disc, en el subdirectori anomenat
TRAÇA. El nom del fitxer en què es guarda la traça correspon a la clau de
l'alumne, i l'extensió pren un valor numèric que serveix per distingir els
fitxers de traça d'un mateix alumne.
Si l'aplicació no troba el directori TRAÇA, el crea. Si té algun problema
d'escriptura (per exemple, que el disc estigui protegit o estigui ple, o que el
Sòcrates s'executi desde un CD-ROM) no escriu cap fitxer de traça.
El fitxer de traça és un fitxer de text que consta de tres apartats.
El primer és  una capçalera amb les dades més significatives de l'exercici:

- Nom i clau de l'alumne.



- Nom dins del disc de la sessió.
- Nom dins del disc de l'exercici.
- Títol de l'exercici.
- Data i hora d'inici.

El segon apartat és un llistat de les operacions efectuades per l'alumne.
Aquestes operacions poden ser:

- Activar una opció d'un menú.
- Fer operacions matemàtiques amb la calculadora.
- Seguir els passos de la maièutica.
- Triar correctament dades o fórmules.

El tercer apartat comença amb la paraula Comentari, i recull tots els
comentaris que l'alumne vulgui fer un cop resolt l'exercici.

Exemple d'arxiu de traça
ALUMNE:   Pere Martí
CLAU:     8c32
SESSIÓ:   C:\SOCRATES\VARIS\VARIS.SES
PROBLEMA: C:\SOCRATES\VARIS\CASA\ CASA.PRB
TÍTOL:    ALÇADA D'UNA CASA
DIA:      15/10/92
HORA:     9:17:33
Necessito més dades.
Necessito més dades.
Necessito més dades.
Necessito més dades.
Necessito més dades.
Necessito més dades.
Tria dades correctes ->1,4,5,6 .
MAIÈUTICA:
Maièutica: Pas 1 Correcte -> si
Maièutica: Pas 2 Correcte -> si
Maièutica: Pas 3 Correcte -> iguals
Maièutica: Pas 4, Intent 1 -> seixanta
Maièutica: Pas 4 Correcte -> 90
Maièutica: Pas 5, Intent 1 -> Iguals
Maièutica: Pas 5, Intent 2 -> Iguals
Maièutica: Pas 5, Correcte -> Semblants.
Maièutica: Pas 6, Intent 1 -> Semblants
Maièutica: Pas 6, Intent 2 -> Relatius
Maièutica: Pas 6, Intent 3 -> A escala
           Es dóna la solució:  Proporcionals.
Maièutica: Pas 7, Intent 1 -> DE
Maièutica: Pas 7 Correcte -> BC
Maièutica: Pas 8 Correcte -> DE

Amagar maièutica.
Calculadora:



24.3
/ 14.2
ET        0

14.2
/ 24.3
= 0.5843621399176955
* 31.1
= 18.1736625514403
MAIÈUTICA:
Maièutica: Pas 9 Correcte -> 18.17
Maièutica: Pas 10 Correcte -> 18.17

COMENTARI:
M'he encallat una mica en els passos 5 i 6, però finalment he
aconseguit trobar la solució de l'exercici.
FI DEL PROBLEMA

3. PROGRAMACIÓ D'EXERCICIS
Els exercicis es programen creant un fitxer de text. Cada un dels punts que
segueixen explica com omplir els diferents apartats d'aquest fitxer,
anomenats camps.
Un camp és una part del fitxer de text, que serveix per donar valor a un o
més paràmetres. El contingut dels diversos camps, que assignen valors als
paràmetres, és el que determina el comportament de l'aplicació.
Els camps estan delimitats per directives, i s'han d'especificar tots
obligadament. Les directives  van precedides d'un caràcter de subratllat.
Aquest caràcter serveix per indicar el començament d'una directiva i, per
tant, no pot utilitzar-se dins del contingut del camp. S'interpreten com a
directiva tots els caràcters que hi ha entre el subratllat i el final de la línia. Hi
ha dos tipus de directives:

Directives simples: consten només d'una paraula especial, escrita en
lletres majúscules, i iniciada per un caràcter de subratllat. Per exemple:

_FI_PROBLEMA.

Directives compostes: estan formades per una directiva simple,
seguida d'un valor que pot ser numèric o no. Per exemple:

_CALCULADORA Cien

Convencions d'escriptura dels formats
En els apartats següents, s'indica el format dels diversos camps. Per
especificar aquest format, s'utilitzen els diferents tipus de text:

Literal : s'ha d'escriure obligadament tal com apareix en el manual. Als
apartats que segueixen, tots els literals estan escrits en lletra cursiva.
Per exemple:

while



No literal, obligatori : entre els caràcters <> s'indica una descripció
del tipus de contingut que s'hi ha de posar. Aquest contingut s'ha
d'escriure sense aquests caràcters. Per exemple, tenim:

<valor_numèric>

i hi escrivim

3

No literal, optatiu : entre els caràcters [] hi ha una descripció del
contingut que s'hi pot posar. Quan el contingut s'escriu, cal ometre
aquests caràcters. Per exemple, tenim:

[Títol de l'exercici]

i hi escrivim

ALÇADA D'UNA CASA

Si després d'un text optatiu hi ha punts suspensius, vol dir que aquest
text es pot repetir tantes vegades com es vulgui. Per exemple:

[nom]
...

Exemple de format:

step <variable> from <inici> to <final> by <pas>
[ordre]
...

end [step ]

I hi escrivim:

step i from 0 to 10
get i+s
put it into s

end step

Preparació de l'exercici
Preparar un exercici consisteix bàsicament a:
- Pensar l'explicació o l'enunciat de l'exercici, i la pregunta que es farà.
- Cercar quines dades es vol que apareguin, i quines són significatives.
Decidir si es poden triar, i si triar-les és optatiu o no.
- Fer els gràfics associats a l'enunciat utilitzant un programa de dibuix.



- Fer un fitxer gràfic on apareguin les fórmules, si és el cas. Decidir quines
fórmules són significatives i si s'han de triar o no. Els fitxers gràfics han de
tenir format BMP o TIF.
- Veure de quina manera es calcula la solució (en cas que s'hagi de calcular)
a partir de les dades.
- Pensar quants intents té l'alumne d'entrar la solució, i els missatges que han
de sortir cada cop que l'alumne introdueixi una resposta incorrecta.
- Preparar la maièutica, incloent els gràfics que hi vagin associats,  les
preguntes, el nombre d'oportunitats de resoldre-les, i els comentaris adients.
- Crear els directoris corresponents a l'exercici.

Fitxer de l'exercici
El fitxer de text que conté l'exercici ha de tenir aquest format:

_PROBLEMA [Títol de l'exercici]
<camps de l'exercici>
_FI_PROBLEMA

La primera directiva és una directiva composta. Tots els caràcters des
d'aquesta directiva fins el final de la línia es prenen com a títol del problema
que apareixerà sobre l'enunciat quan s'executi l'exercici.

Tipus de calculadora
Cada exercici pot tenir o no una calculadora associada. Aquesta calculadora
pot ser  aritmètica o científica. Això es diu amb la directiva composta:

_CALCULADORA <tipus>

on <tipus> pot ser:
Arit  (calculadora aritmètica).
Cien (calculadora científica).
no  (l'exercici no necessita calculadora).

Enunciat
L'enunciat s'especifica entre les dues directives _ENUNCIAT  i
_FI_ENUNCIAT. Pot tenir un  nombre qualsevol de línies, i no ha
d'incloure cap caràcter de subratllat. Les línies poden ser tan llargues com
es vulgui, perquè el programa les ajusta com a paràgrafs dins l'espai
disponible.

_ENUNCIAT
<Text corresponent a l'enunciat>
_FI_ENUNCIAT

Pregunta
El text que apareix en la finestra de pregunta s'especifica entre les directives
_PREGUNTA i _FI_PREGUNTA, i pot tenir les línies que es vulgui. Si el



camp és buit (és a dir, no hi ha cap text entre les dues directives), no apareix
la finestra corresponent en el problema.

_PREGUNTA
[Text de la pregunta]
_FI_PREGUNTA

Cal tenir en compte que la finestra de la pregunta no es pot fer més grossa.
Per tant, cal pensar en una pregunta que s'ajusti a la seva mida.

Gràfics
El format d'aquest camp és:

_GRAFICS
[nom de gràfic]
...
_FI_GRAFICS

Cada text optatiu [nom de gràfic] és una línia de text (acabada amb un salt
de línia), que conté el nom d'un fitxer gràfic, el qual s'incorpora inicialment
a la roda del problema. Els gràfics han d'estar en format BMP o bé TIF.
Exemple:

_GRAFICS
dibu.bmp
c:\art\dibuixos\planell.bmp
_FI_GRAFICS

Nombre de dades inicial
El programa permet especificar un nombre de dades inicial, per si es desitja
que inicialment només presenti un subconjunt de les dades de l'exercici,
perquè l'alumne en vagi demanant més. Això es fa en aquest camp.

_NOMBRE_DADES_PRINCIPI <valor numèric enter>

Valor numèric enter: és el nombre de dades que hi ha inicialment a la
finestra de dades. Si és més gran que el nombre de dades que hi ha, sortiran
totes. Si és 0, no en sortirà cap. Aquestes n dades inicials són les n primeres
dades que s'especifiquin al camp de les dades.
Exemple:

_NOMBRE_DADES_PRINCIPI 4

Dades
El camp de les dades té el format:

_DADES



[ [valor 1] , [valor 2] , [NXS] , [descripció] ]
...
_FI_DADES

Cada una de les línies que hi ha entre les dues directives s'interpreta com
l'especificació d'una dada. Si no hi ha cap línia, la fitxa de dades no
apareixerà a la roda quan s'executi el problema. En cas contrari, apareixerà a
la fitxa la quantitat de dades especificada en el camp anterior, mitjançant la
directiva _NOMBRE _DADES_PRINCIPI.
El text [descripció] és una frase escrita entre cometes, que apareix a la fitxa
de dades com una descripció de la dada.
En el cas que la dada sigui numèrica, es pot dir quantes xifres significatives
es volen considerar. Això es fa amb [NXS].
Els valors [valor 1] i [valor 2] s'especifiquen segons el tipus de valor que
volem que tingui la dada:

Dades numèriques constants: el valor de la dada és el mateix cada
vegada que s'executa un exercici. En aquest cas,  només cal escriure el
valor en el lloc de [valor 1], i deixar buit [valor 2]. Si [NXS]  no és
buit, s'arrodoneix la dada a les xifres significatives que correspongui.
Exemple:

_DADES
40,,, "Pes en kg  del paquet."
3.1416,,3,"Constant PI."
_FI_DADES

Resultats en la finestra de dades:

40 = Pes en kg. del paquet.
3.14 = Constant PI.

Dades numèriques variables: cada cop que s'entra en l'exercici, el
programa calcula un valor aleatori entre [valor 1] i [valor 2]. Cal
especificar [NXS].
Exemple:

_DADES
40, 180, 4, "Distància que recorre l'objecte."
0.0002, 0.04, 3, "Coeficient de fregament."
_FI_DADES

Resultats en la finestra de dades:

64.87 = Distància que recorre l'objecte.
0.0165 = Coeficient de fregament.

Resultat d'una expressió: la dada es calcula en funció d'altres dades



anteriors. [valor 2]  s'ha de deixar en blanc, i  cal especificar [NXS]. A
[valor1] s'introdueix l'expressió corresponent, que va precedida del
caràcter #, i pot utilitzar les operacions aritmètiques habituals, i totes
les funcions descrites més endavant, en l'apartat "Funcions
aritmètiques". A més, es pot utilitzar la funció Dada(), que pren per
paràmetre el nombre d'ordre d'una de les dades anteriors a l'actual dins
el camp, i retorna el valor d'aquesta dada.
Exemple:

_DADES
2.25,,,"Catet A"
5,,,"Catet B"
#Sqrt(5*5+Dada(1)*Dada(1)),,4,"Hipotenusa del triangle."
FI_DADES

Els resultats en la finestra de dades són:

2.25 = Catet A
5 = Catet B
5.482 = Hipotenusa del triangle.

Dades constants no numèriques: la dada s'ha d'escriure en el lloc de
[valor 1], i ha d'anar entre cometes. Tant [valor 2] com [NXS] s'han de
deixar buits.
Exemple:

_DADES
"A",,, "Punt de partida."
"Pitàgores",,,"Autor del teorema que s'ha d'utilitzar."
_FI_DADES

Resultats en la finestra de dades:

A = Punt de partida.
Pitàgores = Autor del teorema que s'ha d'utilitzar.

Dades variables no numèriques: el valor de la dada es pren com un
valor d'entre els possibles. Aquests valors possibles s'escriuen a [valor
1] separats l'un de l'altre per comes, i tancats tots ells entre cometes.
Tant [valor 2] com [NXS] han d'estar buits.
Exemple:

_DADES
"Vermell,Blau,Verd",,,"Color del cotxe."
"Josep,Anna",,,"Persona que baixava l'escala."
_FI_DADES

Resultats en la finestra de dades:



Blau = Color del cotxe.
Anna= Persona que baixava l'escala.

Dades booleanes: aquestes dades són un cas particular de dades no
numèriques, i poden prendre el valor cert o fals. Es tracten de la
manera descrita en els dos punts anteriors. L'aplicació, però, pren com
a sinònims les paraules sí, si, cert, veritat i true, i també les paraules
no, fals, mentida i false. Aquests sinònims també valen si les paraules
estan escrites en majúscules.
Exemple:

_DADES
"SI",,,"El sistema es mou."
"CERT,mentida",,, "El paquet baixa per la rampa."
_FI_DADES

Resultats en la finestra de dades:

Si = El sistema es mou.
No = El paquet baixa per la rampa.

Comentaris: pot ser que el que es desitgi és que aparegui un comentari
sense cap valor associat a la fitxa de dades. En aquest cas, cal deixar
buits [valor 1], [valor 2] i [NXS].
Exemple:

_DADES
,,,"Les mides estan preses en metres."
,,,"No cal considerar el fregament."
_FI_DADES

Resultats en la finestra de dades:

Les mides estan preses en metres.
No cal considerar el fregament.

Dades significatives
Quan es defineix un exercici, es poden definir dades que no són útils per
resoldre'l. D'aquesta manera, s'obliga l'alumne a fer una tria de les dades, per
tal de trobar les que són significatives. Aquesta tria s'ha de fer sempre, però
en l'aplicació Sòcrates, els exercicis poden permetre realitzar-la de forma
explícita. En clicar amb el botó del ratolí sobre les dades de la fitxa de
dades, aquestes poden quedar amb el  color invertit, per indicar que estan
seleccionades. Això s'especifica en el camp de les dades significatives.
Aquest camp té el format:

_DADES_SIGNIFICATIVES <Caràcter>



[llista]
...
_FI_DADES_SIGNIFICATIVES

El text [llista] és una llista de valors numèrics, separats per comes. Cada un
d'aquests valors correspon al número d'ordre d'una dada dins el camp de
dades. Cada llista indica una combinació de dades útil per resoldre el
problema.
El text <Caràcter> pot ser opcionals o obligades. Indica si és obligat o no el
triar les dades significatives abans d'entrar la resposta a l'exercici.
Pel que fa a les dades, hi ha diferents possibilitats:

El problema no té dades
En el cas que el camp de dades estigui buit, la fitxa de dades no sortirà a la
roda. El camp de dades significatives també ha d'estar buit, i es pot posar
opcionals o obligades indistintament com a <Caràcter>.

No s'han de triar les dades
Si el camp de dades no està buit, però no es vol que les dades  es triïn, el
camp de les dades significatives ha d'estar buit, i es pot posar obligades o
opcionals com a <Caràcter>. D'aquesta manera, les dades apareixen dins la
fitxa corresponent, però no es poden seleccionar amb el ratolí.

L'alumne pot triar les dades
Si es vol que l'alumne pugui triar les dades, el camp de dades significatives
ha de contenir les diferents llistes, una per cada combinació vàlida de dades.
El caràcter de les dades, en aquest cas, ha de ser Opcionals.

És obligatori triar les dades
Quan el caràcter de les dades és Obligades, l'aplicació impedeix que
l'alumne entri la solució o posi en marxa la maièutica fins que hagi
seleccionat bé les dades necessàries.
Si el problema admet més d'un mètode de resolució, l'autor pot especificar
més d'una combinació de dades. Qualsevol d'aquestes combinacions serà
acceptada com a vàlida.
Exemple

_DADES_SIGNIFICATIVES Obligades
1,3,5
4,5
_FI_DADES_SIGNIFICATIVES

Fórmules
Per especificar les fórmules d'un exercici fan falta dos elements: la Fitxa de
fórmules i la Fitxa de noms de les fórmules. La Fitxa de fórmules s'ha de
fer amb un programa de dibuix, i enregistrar en un fitxer gràfic, de format
BMP. Es recomana posar una etiqueta (per exemple A:, B:, etc.) al costat de



cada fórmula per tal de poder-s'hi referir posteriorment. Cal tenir en compte
les dimensions de la finestra corresponent, ja que l'aplicació tallarà o afegirà
el que calgui per què el gràfic s'ajusti a la mida de la roda. De qualsevol
manera, la finestra de les fórmules es pot expandir, igual que la resta de
finestres, de manera que es pot visualitzar fins a  una pantalla sencera de
fórmules.
El format del camp és:

_FORMULES <nom del fitxer>
[Títol de la fórmula]
...
_FI_FORMULES

Si no es volen fórmules, s'ha de posar no com a nom del fitxer.
El contingut de la Fitxa de noms de les fórmules s'especifica entre ambdues
paraules clau. Cada línia de text especifica el nom d'una de les fórmules. Per
saber a quina fórmula es refereix, convé que tingui, a més del nom, una
referència per poder relacionar el nom amb la fórmula del gràfic, per
exemple C: Teorema del sinus.  De totes les fórmules que apareixen, també
n'hi poden haver algunes innecessàries per resoldre el problema.

Fórmules significatives
De la mateixa manera que es poden definir quines dades són significatives
per resoldre l'exercici, també es pot fer el mateix amb les fórmules. Quan
l'alumne clica el botó del ratolí sobre la Fitxa de fórmules, apareix la Fitxa
de noms de les fórmules. Dins d'aquesta nova fitxa es pot fer la selecció.
El format d'aquest camp és:

_FORMULES_SIGNIFICATIVES <Caràcter>
[llista]
...
_FI_DADES_SIGNIFICATIVES

Cada [llista] és una llista de valors numèrics, separats per comes. Cada un
d'aquests valors correspon al número d'ordre d'una fórmula dins el camp de
fórmules.
<Caràcter> pot ser Opcionals o Obligades.
Com passa amb les dades, hi ha quatre possibilitats quant a la tria de les
fórmules:

El problema no té fórmules
En aquest cas, el camp de fórmules estarà buit. També ha d'estar-ho el de
fórmules significatives. <Caràcter> pot ser opcionals o obligades.

No s'han de triar les fórmules
En aquest cas, el camp de fórmules significatives ha d'estar buid. <Caràcter>
pot ser opcionals  o obligades indistintament.



L'alumne pot triar-ne
Si el camp de fórmules significatives conté una o més llistes de fórmules i la
paraula de control és Opcionals, l'alumne podrà triar les fórmules sobre la
Fitxa de noms de les fórmules.

És obligatori triar les fórmules
Quan la paraula de control és Obligades, l'aplicació impedeix que l'alumne
entri la solució o que posi en marxa la maièutica fins que hagi seleccionat bé
les fórmules necessàries.
Si el problema admet diversos mètodes de resolució, l'autor pot especificar
més d'una combinació de fórmules. Qualsevol d'aquestes combinacions serà
vàlida.

Solució
El més habitual és que un exercici consisteixi en respondre una pregunta. En
el camp de la solució s'especifica quines respostes s'accepten com a vàlides.
Això es fa per mitjà d'un algorisme programat amb el llenguatge OpenScript
de ToolBook (descrit més endavant). El camp de la solució té el format:

_SOLUCIO <nombre màxim d'intents>
[algorisme]
_FI_SOLUCIO

Si el nombre màxim d'intents que té l'alumne per entrar la solució és zero i
l'alumne vol entrar-la, el programa no li ho permet, i activa la maièutica.
El text [algorisme] és un programa que posa dins de Solucio la resposta
correcta a la pregunta de l'exercici. En cas que la solució sigui numèrica, es
pot posar dins d'NXS el nombre de xifres significatives que cal tenir en
compte quan es compara la resposta correcta amb la resposta entrada per
l'alumne.
A continuació es descriuen els algorismes que corresponen als casos més
freqüents.

La solució consta d'una o més paraules
Si la resposta a la pregunta és una paraula o frase, per exemple Teorema de
Pitàgores, i hi ha tres intents per entrar la solució, s'ha d'escriure:

_SOLUCIO 3
put "Teorema de Pitàgores" into Solucio
_FI_SOLUCIO

La frase que constitueix la resposta s'ha d'escriure entre cometes, i no pot
contenir cap coma. Quan es compara la solució amb la resposta de l'alumne,
l'aplicació passa primer ambdues coses a majúscules, i després les compara
caràcter a caràcter.



La solució pot tenir més d'un enunciat
Això és necessari quan es pot dir la resposta de més d'una manera, o quan hi
ha més d'una resposta correcta. Per exemple, si es pot contestar
indistintament Teorema del Sinus, del sinus o bé Sinus, s'ha d'escriure:

_SOLUCIO 2
put "Teorema del Sinus, del sinus, sinus"  into Solucio
_FI_SOLUCIO

La solució és un valor numèric, enter o real
En el cas que la resposta sigui numèrica, no ha d'anar explicitada entre
cometes. Per acceptar el valor numèric 23,67, per exemple, s'ha de posar:

_SOLUCIO 4
put 4 into NXS
put 23.67 into Solucio
_FI_SOLUCIO

La primera instrucció put serveix per indicar el nombre de xifres
significatives que s'han de tenir en compte en comparar la solució amb la
resposta introduïda per l'alumne. En el cas de l'exemple, aquest nombre seria
quatre, que és el valor per defecte.

La solució pot ser un valor numèric o una frase
També es poden alternar solucions numèriques i no numèriques. En aquest
cas, el nombre de xifres significatives només afecta, naturalment, les
solucions numèriques. Per exemple:

_SOLUCIO 4
put "1/6, un sisè, 0.166" into Solucio
put 3 into NXS
_FI_SOLUCIO

La validació de la resposta de l'alumne es fa seguint els mateixos criteris que
en el cas de la solució numèrica.

La solució es calcula en funció de les dades
Habitualment, això passa quan l'exercici requereix una resposta de tipus
numèric, i s'han definit les dades com a variables (la solució no és la mateixa
cada vegada). En aquest cas, s'ha de posar com a resposta una expressió en
funció de les dades. Per fer-ho, es poden cridar les funcions:

- NumDades() Retorna el nombre de dades del problema.
- Dada(i) Retorna el valor de la dada i-èssima, dins de l'ordre
establert en escriure les dades dins el camp de dades.

A més d'aquestes funcions, es poden usar totes les funcions descrites  en el
capítol Programació d'algorismes. També es pot usar la constant pi.

_SOLUCIO 10



put 5 into NXS
put  3 * Dada(2) - Sin(Dada(1)*pi/180) into Solucio
_FI_SOLUCIO

Missatges d'ajuda
Si l'alumne entra una solució que no és vàlida, es pot especificar quins
missatges li han de sortir, amb el camp  _AJUDA_SOLUCIO. Cada  línia
d'aquest camp, tan llarga com es vulgui, és un missatge d'ajuda. Aquests
missatges van apareixent consecutivament, a mida que l'alumne va fallant.
El format del camp és:

_AJUDA_SOLUCIO
[Missatge]
...
_FI_AJUDA_SOLUCIO

Si no es defineix cap missatge, el sistema en presenta un per defecte, per
informar que la resposta no era correcta. Si hi ha més missatges que intents
d'entrar la solució, el nombre de missatges té preferència sobre la quantitat
d'intents: l'alumne disposa de tants intents com missatges. Hi poden haver
menys missatges que intents. En aquest cas, es repeteix el darrer missatge.
Exemple (hi ha tres intents d'entrar la solució):

_AJUDA_SOLUCIO
Hauries d'utilitzar la tercera dada de la fitxa.
Necessites el fregament per calcular l'acceleració!
_FI_AJUDA_SOLUCIO

Exemple (hem definit deu intents):

_AJUDA_SOLUCIO
La solució no és la correcta. Torna-ho a provar.
_FI_AJUDA_SOLUCIO

En aquest exemple, sortiria el mateix missatge en tots els intents fallats.

Maièutica
Si el problema no té maièutica, cal indicar que el nombre de passos és zero.
En aquest cas, es desactiva l'opció corresponent del menú.
El format del camp de la maièutica és:

_MAIEUTICA <nombre de passos>
_CONTINUAR_SI_ERROR <Si/No>
_DIALEG_CORRECTE <Si/No>
[pas]
...
_FI_MAIEUTICA



Si es respon Si a la directiva _CONTINUAR _SI_ERROR, l'aplicació facilita
la resposta correcta a l'alumne quan acaba els intents d'entrar la solució.
Això es fa si es vol que l'alumne pugui continuar amb l'exercici. Si es respon
No, l'aplicació torna al començament del problema.
Immediatament després d'entrar una resposta correcta, pot sortir un avís que
ho indica, i que resta a la pantalla fins que es prem el botó Vist. Si es vol que
aquest avís surti, cal respondre Sí després de _DIALEG_CORRECTE. En cas
contrari, cal  respondre No.

Exemple:

_MAIEUTICA 6
_CONTINUAR_SI_ERROR No
_DIALEG_CORRECTE Si
<Aquí s'especifiquen els sis passos de la Maièutica>
_FI_MAIEUTICA

Per cada pas, el format és:

_GRAFIC <nom del gràfic>
_PREGUNTA
<Text de la pregunta>
_SOLUCIO <nombre d'intents>
<Algorisme>
_AJUDA
[Missatge]
...

Hi ha quatre directives: _GRAFIC, _PREGUNTA, _SOLUCIO i _AJUDA.
Amb la directiva _GRAFIC es pot associar un fitxer gràfic a la pregunta, de
tal forma que aparegui a la roda quan aquesta es formula. El <nom del
gràfic> ha de ser el nom amb el qual està enregistrat en el disc (per més
explicacions, vegeu l'apartat "Estructura de directoris"). Si no es vol que
aparegui un gràfic, cal respondre no com a nom del fitxer.
El <Text de la pregunta> és la pregunta corresponent a aquell pas de la
Maièutica.
Els text <nombre d'intents >  i  el text <Algorisme> són similars als seus
homònims del camp de la solució global del problema, però s'apliquen al pas
al que pertanyen. Una funció que es pot utilitzar en expressions dins
<Algorisme> és Parcial(i), que retorna el valor de la solució del pas i-èssim
de la maièutica.
El text [Missatge]  és equivalent  al seu homònim del camp d'ajudes per a la
solució.
Exemple:

_GRAFIC no



_PREGUNTA
Quin és el triple de la resposta anterior?

_SOLUCIO 3
put 2 into NXS
put 3* Parcial(5) into Solucio
_AJUDA
Has d'agafar la resposta anterior i multiplicar-la per tres.

Paraules clau
En el camp de paraules clau del fitxer del problema, l'autor pot definir un
text que serveixi d'explicació o aclariment per algunes paraules o frases de
l'enunciat. Quan l'alumne demana ajuda sobre l'aplicació, aquestes paraules
es converteixen en paraules clau, de manera que si es clica a sobre d'elles,
apareix el text explicatiu a la finestra d'ajudes. El format d'aquest camp és:

_PARAULES_CLAU <Número de paraules clau>
[_PARAULA <Inici>,[Fi]
<Text explicatiu>]
...
_FI_PARAULES_CLAU

<Inici> és el número d'ordre de la paraula dins de l'enunciat. Si es posa
també [Fi], es pren com a una sola clau les dues paraules i totes aquelles que
estiguin entre elles.
Exemple:

_PARAULES_CLAU 5
_PARAULA 2

EXPERIMENT:
A la sèrie Probab1, anterior a l'actual, hi trobaràs un exercici

destinat a aquest concepte, repassa'l.
_PARAULA 4,6

TRIAR A L'ATZAR:
Triar sense que es conegui prèviament cap referència del resultat

de la tria, de manera que es pot considerar a tots els resultats possibles
com igualment probables.
_PARAULA 10,11
     BARALLA ESPANYOLA:

Una baralla és un joc complert de cartes.
Una baralla espanyola té 48 cartes, 4 amb el número 1, 4 amb el

número 2, ... ,4 amb el número 9, 4 sotes,4 cavalls i 4 reis.

_PARAULA 17
     ESDEVENIMENT:
     A la sèrie Probab1, anterior a la present, hi trobaràs tres exercicis
destinats a la comprensió d'aquest concepte, llegeix-los.

_PARAULA 20



REI:
En el problema es refereix a cada una de les quatre cartes d'una

baralla espanyola que duen pintada la figura d'un rei.

_FI_PARAULES_CLAU

4. EXEMPLE D'EXERCICI
A continuació es presenta un exemple complet d'exercici. Aquest seria el
contingut del fitxer de text que el defineix.

_PROBLEMA ALÇADA D'UNA CASA

_CALCULADORA Arit

_ENUNCIAT

    Estem al poble de Vallmoll, on hi ha una casa de vuit plantes. És a
la tarda, i el Sol no tardarà a pondre's. Els edificis tenen una ombra
llarga. A la plaça hi ha un monument antic, que és un monòlit prim i
alt.
_FI_ENUNCIAT

_PREGUNTA

     Quant metres d'alçada té l'edifici de vuit plantes?

_FI_PREGUNTA

_GRAFICS
edifici.bmp
edifici2.bmp
_FI_GRAFICS

_NOMBRE_DADES_PRINCIPI 2
_DADES
,,,"Les mides estan preses en metres"
500,1000,4,"Llargada del carrer"
40,90,3,"Amplada del carrer"
13,19,3,"Alçada del monòlit"
19,25,3,"Llargada de l'ombra del monòlit"
26,34,3,"Llargada de l'ombra de l'edifici de vuit pisos"
,,,"Són les sis de la tarda"
_FI_DADES

_DADES_SIGNIFICATIVES Obligades
1,4,5,6
_FI_DADES_SIGNIFICATIVES



_FORMULES no
_FI_FORMULES

_FORMULES_SIGNIFICATIVES Opcionals
_FI_FORMULES_SIGNIFICATIVES

_SOLUCIO 0
    put Dada(6) * Dada(4) / Dada(5) into Solució
_FI_SOLUCIO

_AJUDA_SOLUCIO
_FI_AJUDA_SOLUCIO

_MAIEUTICA 10
_CONTINUAR_SI_ERROR Si
_DIALEG_CORRECTE Si
_GRAFIC no
_PREGUNTA

Hi ha un raig de llum que passa just fregant la punta de l'edifici,
punt D, i cau al terra en el punt F.

Un altre raig passa de la punta del monòlit, punt A, fins al punt C.
Aquests dos rajos són paral·lels?

_SOLUCIO 3
        --1

put "Sí,Si","s" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Si les línies AB i DE són verticals i representen el monòlit i
l'edifici, són paral·leles?
_SOLUCIO 3
        --2

put "Sí,Si","s" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Si tens en compte els dos passos anteriors, còm són els angles A i
D?
_SOLUCIO 3
        --3

put "Iguals" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Si les línies AB i DE són verticals i les línies BC i EF són
horitzontals, quants graus mesuren els angles B i E?
_SOLUCIO 3
        --4

put 90 into Solucio



_AJUDA
_GRAFIC no
_PREGUNTA

Els dos triangles ABC i DEF tenen dos angles iguals: l'angle A és
igual a l'angle D (pas 4), i l'angle B és igual a l'angle E (pas 3).

Com són dos triangles que tenen dos angles iguals?
_SOLUCIO 3
        --5

put "Semblants" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Com tenen els costats dos triangles semblants?
_SOLUCIO 3
        --6

put "Proporcionals" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Tenint en compte el punt anterior, podem dir que si dividim el
costat vertical DE pel costat vertical AB ha de donar igual a dividir el
costat horitzontal EF pel costat x. Quin és el costat x?
_SOLUCIO 3
        --7

put "BC,CB" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Si, tal com s'ha dit en el punt anterior, DE/AB = EF/BC, quin és
l'únic terme que no coneixem d'aquesta igualtat?
_SOLUCIO 3
        --8

put "DE,ED" into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA

Calcula la longitud de DE per mitjà de l'equació vista en el punt
8.
_SOLUCIO 3
        --9

put Dada(6) * Dada(4) / Dada(5) into Solucio
_AJUDA
_GRAFIC no
_PREGUNTA
 Quants metres d'alçada té l'edifici?
_SOLUCIO 3
        --10

put Parcial(9) into Solucio
_AJUDA



_FI_MAIEUTICA

_PARAULES_CLAU 4
_PARAULA 13

     PLANTA:

     Pis, especialment el que és a nivell de carrer, anomenat també
planta baixa o baixos.

     Exemple: La casa on visc té 4 plantes, planta baixa, primer pis,
segon pis i àtic.

_PARAULA 21,23

     PONDRE'S EL SOL:

     Dit del moment de la tarda en què el Sol passa per sota de l'horitzó.

     Exemple: Avui el Sol es pondrà a les set en punt.

_PARAULA 37

     MONUMENT:

     Obra d'escultura o arquitectura erigida per perpetuar el record
d'una persona o d'un fet memorable.

_PARAULA 41

     MONÒLIT:

     Monument de pedra d'una sola peça, generalment estreta i llarga.

_FI_PARAULES_CLAU

_FI_PROBLEMA

5. LA PROGRAMACIÓ DE SESSIONS
El format  que té el fitxer descriptor d'una sessió és el següent:

_SESSIO
_RESOLDRE_EN_ORDRE <Si/No>
_PROBLEMES
<noms dels problemes>
_FI_PROBLEMES



_AJUDA
[Explicació de la sessió]
_FI_AJUDA
_FI_SESSIO
 

Resoldre els exercicis en ordre implica que els menús Triar exercici  i
Exercici següent queden desactivats. Quan això és així, cal procurar que les
sessions no siguin massa llargues, ja que si l'alumne no pot acabar la sessió
no la podrà reprendre sinó que haurà de tornar-la a començar.
Dins del fitxer de la sessió, hi ha el nom dels diferents problemes. El nom
dels diversos problemes s'ha de determinar segons les condicions que
s'expliquen a l'apartat "Estructura de directoris".
A l'apartat _AJUDA hi ha el text que apareix quan es demana informació
sobre la sessió des del menú corresponent.

Exemple de sessió

_SESSIO

_RESOLDRE_EN_ORDRE No
_PROBLEMES
pentagon.prb
casa.prb
assassi.prb
_FI_PROBLEMES

_AJUDA

     Aquesta sessió no té una unitat temàtica sinó que consta d'exercicis
variats que il·lustren el funcionament del programa.

_FI_AJUDA

_FI_SESSIO

6. PROGRAMACIÓ D'ALGORISMES
Aquest apartat explica com programar algorismes per poder calcular
solucions numèriques a partir de les dades, ja sigui en la solució global o en
les solucions dels passos de la maièutica. En la majoria dels casos, els
càlculs són prou simples com per resoldre'ls de manera anàloga als exemples
de l'apartat Programació d'exercicis. El llenguatge de programació que
s'utilitza és l'OpenScript.

L'OpenScript del ToolBook
Les sentències d'aquest llenguatge estan pensades de manera que la sintaxi
correspongui a la de les frases angleses. Per exemple, podem veure la
sentència Put it into Solucio.
Els Scripts (o programes) no són compilats sinó interpretats  pel ToolBook.



Variables
No és obligatori declarar les variables. Malgrat tot, hi ha la possibilitat de
fer-ho posant al principi de l'Script una o més sentències de declaració.
Les sentències de declaració tenen el format:

Local <nom>[,<nom>...]

Els noms de les variables només han de contenir lletres. L'intèrpret ignora les
diferències entre majúscules i minúscules. L'OpenScript no té tipus: totes les
variables són tractades com a tires de caràcters ("strings"). L'única tipificació
que hi ha consisteix a identificar tires de caràcters de format especial. Aquest
format especials pot ser:

Format numèric enter.
Format numèric real.
Format booleà (TRUE o FALSE).
Format de data (date).
Format d'hora (time).

Variables especials
L'OpenScript permet usar diferents variables especials, que ja estan
predefinides. Una d'elles és la variable it. Aquesta variable no té cap
significat específic, i serveix per donar un aspecte de "llenguatge natural" a
les sentències. L'aplicació Sòcrates en defineix dues més:
- Solucio: variable on s'ha de deixar la resposta correcta, numèrica o no, a la
pregunta efectuada.
- NXS: s'hi ha de posar el nombre de xifres significatives amb què volem
comparar la resposta correcta amb la que ha donat l'alumne. Si la resposta no
és numèrica, el contingut d'aquesta variable s'ignora.

Operadors
Aritmètics

+ - * /
^  (exponenciació)
div  (divisió entera)
mod  (mòdul de la divisió entera)

Lògics
De comparació:

<      <=       <>      =      >      >=
is      is not.

D'operands lògics:
and     or     not.

D'inclusió de tires de caràcters:
is in     is not in     contains.



De caràcters (strings)
&, &&

Els dos operands són de concatenació. La diferència és que &&  deixa un
espai entre les tires de caràcters concatenades i &,  no.

Comandes
Assignació
Sintaxi:

put <expressió> into <variable>
put <expressió> before <variable>
put <expressió> after <variable>

Si s'utilitza la primera forma, el resultat d'avaluar l'expressió (que pot
contenir funcions i operadors) es sitúa dins de la variable, i elimina el que hi
havia abans.
Si s'utilitza la forma before  o after, el resultat d'avaluar l'expressió es
concatena amb el contingut de la variable per l'esquerra o la dreta
respectivament.
Per exemple:

put "No s'hi val" into LaMevaVariable
put "Què tal? " into LaMevaVariable
put "Hola. " before LaMevaVariable
put "Jo estic bé, gràcies." after LaMevaVariable

El valor de la variable LaMevaVariable després d'aquestes operacions és
"Hola. Què tal? Jo estic bé, gràcies."
Hi ha l'ordre especial get, que posa un valor dins de la variable It:

get "Hola. Què tal? Jo estic bé, gràcies."

Aquesta ordre faria el mateix que la de l'exemple anterior, però amb la
variable It..

Altres ordres
- Increment <variable>
- Decrement <variable>

Augmenten i disminueixen, respectivament, el valor d'una variable en una
unitat.

Funcions aritmètiques
Les funcions aritmètiques disponibles són:

- abs():  valor absolut.
- acos(): arc cosinus.



- annuityFactor(): pren per paràmetres el rati d'interès per període en tant per
u, i el nombre de períodes. Retorna el valor calculat per la fórmula (1-
(1+<rati>)^(-<períodes)))/<rati>.
- asin(): arc sinus.
- atan(): arc tangent.
- atan2(): pren per paràmetres el catet oposat i el catet contigu a l'angle.
- average(): pren per paràmetres una llista de valors separats per comes, i en
calcula la mitjana aritmètica.
- ceiling(): retorna l'enter immediatament superior al nombre donat.
- compoundFactor(): calcula l'interès compost. Pren per paràmetres l'interès
en tant per u i el nombre de períodes.
- cos(): cosinus.
- cosh(): cosinus hiperbòlic.
- exp(): retorna el nombre e elevat al valor del paràmetre.
- floor(): retorna l'enter immediatament inferior al nombre donat.
- hypotenuse(): rep per paràmetres els dos catets d'un triangle rectangle, i en
retorna l'hipotenusa.
- ln(): logaritme neperià.
- log(): logaritme decimal.
- max(): retorna el valor més gran d'una llista de nombres separats per comes.
- min(): retorna el valor inferior d'una llista de nombres.
- random(): retorna un enter entre 1 i l'enter especificat, ambdós inclosos. Per
iniciar la successió de nombres aleatoris, s'usa la comanda seed n, on n és un
enter.
- round(): arrodoneix un nombre a l'enter més pròxim.
- sin(): sinus.
- sinh(): sinus hiperbòlic.
- sqrt(): arrel quadrada.
- sum(): retorna la suma d'una llista de nombres.
- tan(): Tangent.
- tanh(): tangent hiperbòlica.
- truncate(): retorna el valor truncat del paràmetre.

Observació: El valor dels angles s'ha d'expressar en radians.

Estructures de control

if/then/else

if <expressió booleana>
[then] <ordres>
[else <ordres>]

end [if]

conditions/when/else

conditions



[when <expressió booleana>
[<ordres>]]

...
[else

<ordres>]
end [conditions]

while

while <expressió booleana>
<ordres>

end [while]

do/until

do
<ordres>

until <expressió booleana>

step

step <variable> from <inici> to <final> [by <pas>]
<ordres>

end [step]

Sentència equivalent al for/to  del Pascal. Els passos poden ser positius o
negatius.

Altres aspectes de l'OpenScript

Comentaris
Els comentaris van precedits per una ratlla discontínua --, i s'acaben a la
mateixa línia que comencen.

Línies massa llargues
L'intèrpret de l'OpenScript no té problemes amb les línies molt llargues, però
si es vol fer el programa més llegible, es pot partir la línia amb el caràcter
"\".
Per exemple:

put var1 + var2 -( var2 * var3 \
(var4 + var1)) into Solucio

L'única cosa que cal tenir en compte és que no es pot partir una tira de
caràcters, d'aquesta manera:



get "La veritat és que això \
no és massa correcte"

S'hauria de fer comptant amb l'operador & :

get "Així és com "& \
"s'ha de fer."

Cometes dins de línies de caràcters
Quan es vol incloure cometes dins d'una línia de caràcters, s'han de posar
cometes dobles. Per exemple:

Expressió      Contingut de la variable

"He vist una ""bici"" aquí." He vist una "bici" aquí.
"""Hola""" "Hola"
"""" "

Les llistes d'elements
El llenguatge OpenScript no té vectors (arrays de Pascal). Per suplir-los, es
pot usar la paraula item. L'intèrpret permet definir una llista d'elements
separats per comes. Per tal d'accedir a l'element i-èssim, utilitzem item i. Per
exemple:

put "A,B,C,D,E" into Llista
put item 2 of Llista into Element

Ara, Element té el valor "B". També es pot utilitzar a la inversa:

put "A,B,C,D,E" into Llista
put "X" into item 3 of Llista

Ara, Llista té el valor "A,B,X,D,E".

Exemples de càlcul de solucions

Exemple 1:
_SOLUCIO 3

-- Calcula la mitjana aritmètica de les dades.
-- Utilitza cinc xifres significatives.
Local i,suma -- No és necessari declarar-les.

put 0 into suma
step i from 1 to NumDades()

put suma +Dada(i) into suma
end
put suma/NumDades() into Solucio
put 5 into NXS



_FI_SOLUCIO

Exemple 2:
_SOLUCIO 3
-- Calcula la suma dels valors absoluts de les dades.
-- Utilitza quatre xifres significatives.

Local i,suma -- No és necessari declarar-les.

put 0 into suma
put 1 into i
while i =< NumDades()

if Dada(i) > 0
put suma + Dada(i) into suma

else
put suma - Dada(i) into suma

end
Increment i

end
put suma into Solucio
put 4 into NXS

_FI_SOLUCIO

Exemple 3:
_SOLUCIO 3
-- Calcula la suma de les dades positives, la suma
-- aritmètica en valor absolut de les dades negatives,
-- i pren com a solució la mitjana aritmètica calculada
-- a partir de la més gran de les dues sumes.
-- Utilitza cinc xifres significatives.

Local i,suma1,suma2,mitjana1,mitjana2

put 0 into suma1
put 0 into suma2
put 1 into i
repeat

conditions
when Dada(i) = 0
-- No fem res, ja que la dada és nul·la.
when Dada(i) > 0
-- La dada és positiva

put suma1 + Dada(i) into suma1
else
-- La dada no és zero ni positiva (és
-- negativa.).

put suma2 - Dada(i) into suma2
end
Increment i

until i > NumDades()
if suma1 > suma2



put suma1/NumDades() into Solucio
else

put suma2/NumDades() into Solucio
end
put 5 into NXS

_FI_SOLUCIO


